

DEPARTMENT OF THE INTERIOR

U.S. GEOLOGICAL SURVEY

Debye temperatures of selected silicate minerals

by

Richard A. Robie
U.S. Geological Survey
Reston, VA 22092¹

Open-File Report 88-663

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards.

¹USGS duty station

ABSTRACT

The acoustic Debye temperatures, θ_D^e , for 31 silicate minerals have been calculated from their elastic stiffness (c_{ij} 's) constants and/or their heat capacities at very low temperatures. Calculation of the θ_D^e values from the c_{ij} 's were done on a Hewlett-Packard series 200 computer using a program written in Hewlett-Packard BASIC.

For those silicates having no magnetic or electronic contributions to their heat capacity for which modern heat capacity measurements below 20 K exist, the calorimetric Debye temperatures, θ_D^c , are in good agreement with θ_D^e values obtained from the elastic constants.

INTRODUCTION

Debye's (1912) theory for the heat capacity of a solid provides a simple one parameter model which correctly describes the temperature dependence of C_v at low temperatures. It is based upon a model for the frequency spectrum, $G(\nu)$, of a solid whose significant features are a cutoff or maximum frequency ν_{\max} , a total of $3N_A$ frequencies where N_A is Avogadro's constant, and a quadratic variation of the number of frequencies (i.e., $dn/d\nu \sim \nu^2$) at low frequency. Debye's model spectrum is given by (e.g. Zemansky, 1968)

$$G(\nu) = (9N_A/\nu_{\max}^3)\nu^2 \quad (1)$$

The Debye temperature, θ_D^e , is directly related to this maximum frequency by

$$\theta_D^e = (h/k)\nu_{\max} \quad (2)$$

or, to the mean velocity of sound in the crystal by

$$\theta_D^e = (h/k)[3N\rho/4\pi M]^{1/3}\nu_m \quad (3)$$

where h is Planck's constant = 6.626076×10^{-34} J·s, k is the Boltzmann constant = 1.38066×10^{-23} J·K⁻¹, Avogadro's constant, $N_A = 6.0221367 \times 10^{23}$ mol⁻¹, ρ is the density, M the formula weight, and ν_m is the mean velocity of sound.

Because of the simplicity of the theory and its considerable success in calculating the heat capacity at low temperatures, the concept of a characteristic or Debye temperature has become a central concept in the equation of state of solids. Thus the Debye temperature enters into the Grüneisen theory of thermal expansion, the Lindemann melting relation, and as pointed out by Anderson (1965), given the Debye temperature one may estimate fairly closely the velocity of shear waves in a solid from the approximate relation

$$v_s \approx (\theta_D/280)(M/q\rho)^{1/3} \text{ km sec}^{-1} \quad (4)$$

where q is the number of atoms in the chemical formula. Thus values for the Debye temperature are of some use to the geophysicist trying to model the thermal parameters of mantle minerals.

θ_D FROM $C_p^o(T)$ MEASUREMENTS

At sufficiently low temperatures the vibrational spectrum of a solid is composed of frequencies in the "acoustic" range (i.e., $5 \times 10^{13} \text{ sec}^{-1}$ or less), optical frequencies are not excited and the heat capacity is given by (e.g., Slater, 1939)

$$C = \frac{12R\pi^4}{5} \left(\frac{T}{\theta_D} \right)^3 = 1943.8 \left(\frac{T}{\theta_D} \right)^3 \quad (5)$$

where $R = N_A \cdot k$ is the gas constant, $= 8.31451(7) \text{ J} \cdot \text{mol}^{-1} \cdot \text{K}^{-1}$, T is the Kelvin temperature, and $C = C_v \approx C_p^o$. Thus C will vary as T^3 and a plot of C/T against T^2 will be a straight line passing through the origin. One may therefore obtain a purely acoustical value for the Debye temperature (θ_D^C) from heat capacity measurements at low temperatures using (5). We stress that equation (5) is valid only for temperatures of the order of $T < \theta_D/50$. This is shown in Figure 1 where we have plotted the C_p^o measurements of Krupka et al. (1985) for synthetic enstatite and wollastonite, and those of Hemingway et al. (1986) for phenakite.

Equation (3) and (5) are applicable to monoatomic solids only. For a polyatomic solid one must multiply (3) by $q^{1/3}$ and (5) by q .

θ_D FROM ELASTIC CONSTANTS

One may also obtain a (purely acoustical) value for θ_D from the mean velocity of sound in a crystal from equation (3). The difficulty in calculating a mean sound velocity in a low-symmetry crystal arises from the complex variation of v with direction as shown in Figure 2. For any direction in a crystal the velocity of sound can be calculated from the elastic stiffness constants and the density of the crystal by using the Christoffel equation

$$\begin{vmatrix} \Gamma_{11} - \rho v^2 & \Gamma_{12} & \Gamma_{13} \\ \Gamma_{12} & \Gamma_{22} - \rho v^2 & \Gamma_{23} \\ \Gamma_{13} & \Gamma_{23} & \Gamma_{33} - \rho v^2 \end{vmatrix} = 0, \quad (6)$$

which is a cubic in ρv^2 . The Γ_{ij} are complex functions of the elastic constants and of the direction cosines of the propagation direction of the sound wave. For any direction in the crystal there will be three solutions to this equation corresponding to the two shear (transverse) velocities and one longitudinal (compressional) velocity. By solving the Christoffel equation for various directions one may, by a numerical integration procedure, obtain the mean sound velocity. The calculational procedures are discussed in detail by Robie and Edwards (1966). Thus if we have values for the c_{ij} for a crystal, we can obtain a purely acoustical value for the Debye temperature. To this end, the procedure described by Robie and Edwards (1966) was programmed in Hewlett-Packard BASIC for a personal computer. A listing of this program is given in the Appendix.

The development of the Brillouin scattering technique for measuring the velocity of sound in microcrystals by Weidner et al. (1975) together with

the velocity inversion procedure to obtain the elastic stiffness constants by Weidner and Carleton (1977) was a major experimental advance and in recent years has provided accurate elastic constant data for a number of low-symmetry silicate minerals.

Because of the usefulness of the Debye model for extrapolating measured C_p° data to zero kelvin, we have collected the more recent elastic constant data for silicate minerals and used these data to calculate Debye temperatures. To be strictly rigorous one should compare the θ_D values obtained from low-temperature heat capacity data with θ_D calculated from elastic constants *also* measured at low temperatures. The temperature variation of the elastic constants is usually fairly small and can normally be neglected. As an example the Debye temperatures of forsterite calculated from the elastic constants of Sumino et al. (1977) measured at 83 and 293 K are 767 and 757 K respectively.

The results of these calculations together with the source of the elastic constants are listed in Table 1. Where available, we have also tabulated values for θ_D^C calculated from low temperature heat capacity data in the temperature range $T < \theta/25$.

Table 1. Debye temperatures of some silicate minerals calculated from their elastic stiffness constants or from low-temperature heat capacities.

Name and formula	Density g·cm ⁻³	v _m km·sec ⁻¹	θ _D ^e		θ _D ^c K	Source
			K	K		
Coesite SiO ₂	2.911	5.042	666			Weidner and Carleton (1977)
Stishovite SiO ₂	4.290	7.864	1183			Weidner, Bass Ringwood and Sinclair (1982)
Kyanite Al ₂ SiO ₅	3.675			1100		Robie and Hemingway (1984)
Andalusite Al ₂ SiO ₅	3.145	6.204	838	855		Vaughan and Weidner (1978) Robie and Hemingway (1984)
Phenakite Be ₂ SiO ₄	2.960	6.473	933	939		Yeganeh-Haeri and Weidner (1985) Hemingway et al. (1986)
Beryl Be ₃ Al ₂ (Si ₆ O ₁₈)	2.698	6.043	799			Yoon and Newnham (1973)
Tourmaline (uvite) CaMg ₃ (MgAl ₅)BSi ₆ O ₃₀ (OH,F)	3.061	5.673	770			Tatli (1985)
Sillimanite Al ₂ SiO ₅	3.241	5.927	809	730		Vaughan and Weidman (1978) Robie and Hemingway (1984)
Grossular Ca ₃ Al ₂ Si ₃ O ₁₂	3.595	5.981	816			Halleck (1973)
Uvarovite Ca ₃ Cr ₂ Si ₃ O ₁₂	3.850	5.440	733			Bass (1986)
Andradite Ca ₃ Fe ₂ Si ₃ O ₁₂	3.836	5.386	721			Bass (1986)

Table 1 (page 2)

Name and formula	Density g·cm ⁻³	v _m km·sec ⁻¹	θ _D ^e		θ _D ^c		Source
			K	K	K	K	
Pyrope Mg ₃ Al ₂ Si ₃ O ₁₂	3.563	5.589	789				Leitner, Weidner and Libermann (1980)
Cobalt olivine Co ₂ SiO ₄	4.702	4.039	548				Sumino (1979)
Fayalite Fe ₂ SiO ₄	4.397	3.786	507				Sumino (1979)
Forsterite Mg ₂ SiO ₄	3.214	5.553	758,756		774		Graham and Barsch (1969), Sumino et al. (1977) Robie et al. (1982)
Tephroite Mn ₂ SiO ₄	4.128	4.050	533				Sumino (1979)
Liebsbergite Ni ₂ SiO ₄	4.933	4.464	615				Bass et al. (1984)
Spinel Ni ₂ SiO ₄	5.351	4.997	708				Bass et al. (1984)
Zircon ZrSiO ₄	4.649	5.351	718				Ozkan, Cartz and Jamieson (1974)
Wollastonite CaSiO ₃	2.909				528		Krupka et al. (1985)
Ca-Al pyroxene CaAl ₂ SiO ₆	3.435				820		Haselton, Hemingway and Robie (1983)
Hedenbergite CaFeSi ₂ O ₆	3.657	4.564	619				Kandelin and Weidner (1988)

Table 1 (page 3)

Name and formula	Density g·cm ⁻³	v _m km·sec ⁻¹	θ _D ^e		Source
			K	K	
Diopside CaMgSi ₂ O ₆	3.277	4.989	668	580-650	Levien, Weidner and Prewitt (1979) Leadbetter et al. (1977)
Enstatite MgSiO ₃	3.194	5.376	732		Weidner, Wang and Ito (1978)
Bronzite Mg _{0.96} Fe _{0.06} SiO ₃	3.272	5.25	717		Duffy and Vaughan (1988)
Bronzite Mg _{0.8} Fe _{0.2} SiO ₃	3.354	5.209	708		Frisillo and Barsch (1972)
Orthoferrosillite FeSiO ₃	4.002	4.026	540		Bass and Weidner (1984)
Jadeite NaAlSi ₂ O ₆	3.33	5.578	768		Kandelin and Weidner (1988)
Muscovite KAl ₂ [AlSi ₃ O ₁₀](OH) ₂	2.844	3.591	480		Vaughan and Guggenheim (1986)
Danburite CaB ₂ Si ₂ O ₈	3.004	5.307	723		Ozkan and Cartz (1986)
Nepheline KNa ₃ Al ₄ Si ₄ O ₁₆	2.571	3.961	500		Bonczar and Barsch (1975)

REFERENCES

- Anderson, O.L., 1965, An approximate method of estimating shear velocity from specific heat. *Journal of Geophysical Research*, v. 70, p. 4726-4728.
- Bass, J.D., 1986, Elasticity of uvarovite and andradite garnets. *Journal of Geophysical Research*, v. 91, p. 7505-7516.
- Bass, J.D. and Weidner, D.J., 1984, Elasticity of single-crystal orthoferrosilite. *Journal of Geophysical Research*, v. 89, p. 4359-4371.
- Bass, J.D., Weidner, D.J., Hamaya, N., Ozima, M. and Akimoto, S., 1984, Elasticity of the olivine and spinel polymorphs of Ni_2SiO_4 . *Physics and Chemistry of Minerals*, v. 10, p. 261-272.
- Bonczar, L.J. and Barsch, G.R., 1975, Elastic and thermoelastic constants of nepheline. *Journal of Applied Physics*, v. 46, p. 4339-4340.
- Debye, P., 1912, Zur theorie der spezifischen Warmen. *Annalen der Physik*, v. 39, p. 789-839.
- Duffy, T.S. and Vaughan, M.T., 1988, Elasticity of enstatite and its relationship to crystal structure. *Journal of Geophysical Research*, v. 93, p. 383-391.
- Frisillo, A.L. and Barsch, G.R., 1972, Measurement of single-crystal elastic constants of bronzite as a function of temperature and pressure. *Journal of Geophysical Research*, v. 77, p. 6360-6384.
- Graham, E.K. and Barsch, G.R., 1969, Elastic constants of single-crystal forsterite as a function of temperature and pressure. *Journal of Geophysical Research*, v. 74, p. 5949-5960.
- Halleck, P.M., 1973, The compression and compressibility of grossularite garnet: a comparison of x-ray and ultrasonic methods. Ph.D. Thesis, University of Chicago, Chicago, Illinois.
- Haselton, H.T., Jr., Hemingway, B.S. and Robie, R.A., 1983, Low-temperature heat capacities of $\text{CaAl}_2\text{SiO}_6$ glass and pyroxene and thermal expansion of $\text{CaAl}_2\text{SiO}_6$ pyroxene. *American Mineralogist*, v. 69, p. 481-489.

- Hemingway, B.S., Barton, M.D., Robie, R.A. and Haselton, H.T., Jr., 1986, The heat capacity and thermodynamic functions for beryl, $\text{Be}_3\text{Al}_2\text{Si}_6\text{O}_{18}$, phenakite, Be_2SiO_4 , euclase, $\text{BeAlSi}_4(\text{OH})$, bertrandite, $\text{Be}_4\text{Si}_2\text{O}_7(\text{OH})_4$, and chrysoberyl. *American Mineralogist*, v. 71, p. 557-568.
- Kandelin, J. and Weidner, D.J., 1988, The single-crystal elastic properties of jadeite. *Physics of the Earth and Planetary Interiors*, v. 50, p. 251-260.
- Kandelin, J. and Weidner, D.J., 1988, Elastic properties of hedenbergite. *Journal of Geophysical Research*, v. 93, p. 1063-1072.
- Krupka, K.M., Robie, R.A., Hemingway, B.S., Kerrick, D.M. and Ito, J., 1985, Low-temperature heat capacities and derived thermodynamic properties of anthophyllite, diopside, enstatite, and wollastonite. *American Mineralogist*, v. 70, p. 249-260.
- Leadbetter, A.J., Jeapes, A.P., Waterfield, C.G. and Wycherley, K.E., 1977, The low temperature heat capacity of some glass ceramics. *Chemical Physics Letters*, v. 52, p. 469-471.
- Leitner, B.J., Weidner, D.J. and Liebermann, R.C., 1980, Elasticity of single crystal pyrope and implications for garnet solid solution series. *Physics of the Earth and Planetary Interiors*, v. 22, p. 111-121.
- Levien, L., Weidner, D.J. and Prewitt, C.T., 1979, Elasticity of diopside. *Physics and Chemistry of Minerals*, v. 4, p. 105-113.
- Ozkan, H., Cartz, L., and Jamieson, J.C., 1974, Elastic constants of nonmetamict zirconium silicate. *Journal of Applied Physics*, v. 45, p. 556-562.
- Ozkan, H. and Cartz, L., 1986, Elastic constants of calcium borosilicate ($\text{CaB}_2\text{Si}_2\text{O}_8$). *High Temperatures - High Pressures*, v. 18, p. 675-678.
- Robie, R.A. and Edwards, J.L., 1966, Some Debye temperatures from single-crystal elastic constant data. *Journal of Applied Physics*, v. 37, p. 2659-2663.

- Robie, R.A., Hemingway, B.S. and Takei, H., 1982, Heat capacities and entropies of Mg_2SiO_4 , Mn_2SiO_4 , and Co_2SiO_4 between 5 and 380 K. *American Mineralogist*, v. 67, p. 470-482.
- Robie, R.A. and Hemingway, B.S., 1984, Entropies of kyanite, andalusite, and sillimanite: additional constraints on the pressure and temperature of the Al_2SiO_5 triple point. *American Mineralogist*, v. 69, p. 298-306.
- Slater, J.C., 1939, *Introduction to Chemical Physics*. New York, McGraw-Hill, 521 p.
- Sumino, Y., 1979, The elastic constants of Mn_2SiO_4 , Fe_2SiO_4 and Co_2SiO_4 and the elastic properties of olivine group minerals at high temperature. *Journal of the Physics of the Earth*, v. 27, p. 209-238.
- Sumino, Y., Nishizawa, O., Goto, T., Ohono, I. and Ozima, M., 1977, Temperature variation of elastic constants of single crystal forsterite between -190° and 400°C . *Journal of the Physics of the Earth*, v. 25, p. 377-392.
- Tatli, A., 1985, Zero-field elastic constants of uvite. *Journal of Physics and Chemistry of Solids*, v. 46, p. 1015-1018.
- Vaughan, M.T. and Weidner, D.J., 1978, The relationship of elasticity and crystal structure in andalusite and sillimanite. *Physics and Chemistry of Minerals*, v. 3, p. 133-144.
- Vaughan, M.T. and Guggenheim, S., 1986, Elasticity of muscovite and its relationship to crystal structure. *Journal of Geophysical Research*, v. 91, p. 4657-4664.
- Weidner, D.J. and Carleton, H.R., 1977, Elasticity of coesite. *Journal of Geophysical Research*, v. 82, p. 1334-1346.
- Weidner, D.J., Swyler, K. and Carleton, H.R., 1975, Elasticity of micro-crystals. *Geophysical Research Letters*, v. 2, p. 189-192.
- Weidner, D.J., Wang, H. and Ito, J., 1978, Elasticity of orthoenstatite. *Physics of the Earth and Planetary Interiors*, v. 17, p. 7-13.

- Weidner, D.J., Bass, J.D., Rigwood, A.E. and Sinclair, W., 1982, The single-crystal elastic module of stishovite. *Journal of Geophysical Research*, v. 87, p. 4740-4746.
- Yeganeh-Haeri, A. and Weidner, D.J., 1985, Single crystal elastic moduli of a beryllium silicate: phenacite (Be_2SiO_4). *EOS*, v. 66, p. 358.
- Yoon, H.S. and Newnham, R.E., 1973, The elastic properties of beryl. *Acta Crystallographica*, v. A29, p. 507-509.
- Zemansky, M.W., 1968, *Heat and Thermodynamics (Fifth Edition)*. New York, McGraw-Hill, 657 p.

FIGURE CAPTIONS

Figure 1. Low-temperature molar heat capacities for enstatite (squares) and wollastonite (triangles) from Krupka et al. (1985) and for phenakite (diamonds) from Hemingway et al. (1986) plotted as C_p°/T against T^2 for $T < \theta_D/35$. Solid straight line correspond to calculated values using $\theta_D = 939$ K for phenakite, 692 K for enstatite, and 539 K for wollastonite.

Figure 2. Velocity of sound in muscovite in km/sec in the (010) plane calculated from the elastic constants of Vaughan and Guggenheim (1986).




Figure 1




Figure 2

APPENDIX

A computer program was written in BASIC to calculate the mean velocity of sound given values for the single crystal elastic constants using the relations given by Robie and Edwards (1966). The program was written for a Hewlett-Packard 9826 computer. A listing of the computer program is given as Table 2.

In response to the request by the computer, the user supplies the name, crystal density, crystal system, number of atoms in the formula, indicates whether the elastic constants are the stiffnesses (c_{ij}) or compliances (s_{ij}) and the initial, final and incremental values for the azimuthal angle ϕ , and similarly for the zenith angle θ . The computer then solves for the 3 roots of equation (6) at each of the specified values of θ and ϕ and numerically evaluates the following expression

$$3/v_m^3 = \int_0^{4\pi} [(1/v_1^3) + (1/v_2^3) + (1/v_3^3)](d\Omega/4\pi), \quad (7)$$

where v_1 , v_2 , and v_3 are the roots of Eq. (5) and $d\Omega$ is the differential solid angle

$$d\Omega = \sin\theta d\theta d\phi. \quad (8)$$

to obtain the mean velocity of sound, and hence the Debye temperature.

The program lists the values for the sound velocity (two shear wave velocities and one longitudinal velocity) at each specified θ and ϕ . It can also generate a file of these velocities on a floppy disk for use with a subsidiary plotting program.


```

10 REM CRYSTAL WAVE VELOCITIES AND DEBYE TEMPS REVISED 12/2/1988
20 DIM C(6,6),S(6,6),G(3,3),V(3),A1(73,73),A2(73,73),A3(73,73),A4(73,73),P1$[
14],Name$[60]
30 PRINT "NAME AND OTHER INFORMATION?"
40 INPUT Name$
50 PRINT "DATE MM/DD/YY"
60 INPUT Day$
70 PRINT "CRYSTAL SYSTEM CAN ONLY HAVE THE NAMES: CUBIC, TET, HEX, ORTH, MON,
TRIG"
80 PRINT "CRYSTAL SYSTEM?"
90 INPUT System$
100 P1$="2(3D),3(2D.3D)"
110 Avog=6.022094E+23
120 Hk=4.799281E-11
130 REM PRINT "IF THE ELASTIC CONSTANTS ARE S(I,J) AND NOT C(I,J) ENTER 1"
140 REM INPUT Jjjj
150 PRINT "DENSITY?"
160 INPUT Rho
170 PRINT "NUMBER OF ATOMS IN FORMULA?"
180 INPUT Iq
190 PRINT "ENTER 1 TO GENERATE A PLOTTING FILE OR ENTER 0 TO CALCULATE DEBYE T
"
200 INPUT Yyyy
210 PRINT "IF YOU WISH TO PLOT INSERT A FORMATTED DISK AND PRESS CONTINUE"
220 PAUSE
230 IF Yyyy=0 THEN 310
240 Tmin=0
250 Tm=360
260 Dt=5
270 Pmin=0
280 Pm=360
290 Dp=5
300 GOTO 450
310 PRINT "FORMULA WEIGHT, IN GRAMS"
320 INPUT Formwt
330 PRINT "THETA MINIMUM ?"
340 INPUT Tmin
350 PRINT "THETA MAXIMUM ?"
360 INPUT Tm
370 PRINT "THETA INCREMENT ?"
380 INPUT Dt
390 PRINT "PHI MINIMUM ?"
400 INPUT Pmin
410 PRINT "PHI MAXIMUM ?"
420 INPUT Pm
430 PRINT "PHI INCREMENT ?"
440 INPUT Dp
450 Sy=360./Pm*180./Tm
460 St=0
470 Sl=0
480 Sm=0
490 P=.01745329
500 Dtt=Dt
510 Dt=Dt*P
520 Dpp=Dp
530 Dp=Dp*P
540 IF System$="CUBIC" THEN 560
550 IF System$<>"CUBIC" THEN 690
560 PRINT "C(1,1) ?"
570 INPUT C(1,1)
580 PRINT "C(1,2) ?"
590 INPUT C(1,2)

```

```

600 PRINT "C(4,4) ?"
610 INPUT C(4,4)
620 C(2,2)=C(1,1)
630 C(3,3)=C(2,2)
640 C(5,5)=C(4,4)
650 C(6,6)=C(5,5)
660 C(1,3)=C(1,2)
670 C(2,3)=C(1,3)
680 GOTO 1830
690 IF System$="ORTH" THEN 710
700 IF System$<>"ORTH" THEN 900
710 PRINT "C(1,1) ?"
720 INPUT C(1,1)
730 PRINT "C(2,2) ?"
740 INPUT C(2,2)
750 PRINT "C(3,3) ?"
760 INPUT C(3,3)
770 PRINT "C(4,4) ?"
780 INPUT C(4,4)
790 PRINT "C(5,5) ?"
800 INPUT C(5,5)
810 PRINT "C(6,6) ?"
820 INPUT C(6,6)
830 PRINT "C(1,2) ?"
840 INPUT C(1,2)
850 PRINT "C(1,3) ?"
860 INPUT C(1,3)
870 PRINT "C(2,3) ?"
880 INPUT C(2,3)
890 GOTO 1830
900 IF System$="TET" THEN 920
910 IF System$<>"TET" THEN 1110
920 PRINT "C(1,1) ?"
930 INPUT C(1,1)
940 PRINT "C(3,3) ?"
950 INPUT C(3,3)
960 PRINT "C(4,4) ?"
970 INPUT C(4,4)
980 PRINT "C(6,6) ?"
990 INPUT C(6,6)
1000 PRINT "C(1,2) ?"
1010 INPUT C(1,2)
1020 PRINT "C(1,3) ?"
1030 INPUT C(1,3)
1040 PRINT "C(1,6) ?"
1050 INPUT C(1,6)
1060 C(2,2)=C(1,1)
1070 C(2,3)=C(1,3)
1080 C(5,5)=C(4,4)
1090 C(2,6)=-C(1,6)
1100 GOTO 1830
1110 IF System$="HEX" THEN 1130
1120 IF System$<>"HEX" THEN 1280
1130 PRINT "C(1,1)"
1140 INPUT C(1,1)
1150 PRINT "C(3,3) ?"
1160 INPUT C(3,3)
1170 PRINT "C(4,4) ?"
1180 INPUT C(4,4)

```

```

1190 PRINT "C(1,2) ?"
1200 INPUT C(1,2)
1210 PRINT "C(1,3) ?"
1220 INPUT C(1,3)
1230 C(2,2)=C(1,1)
1240 C(2,3)=C(1,3)
1250 C(5,5)=C(4,4)
1260 C(6,6)=1/2*(C(1,1)-C(1,2))
1270 GOTO 1830
1280 IF System$="TRIG" THEN 1300
1290 IF System$<>"TRIG" THEN 1530
1300 PRINT "C(1,1) ?"
1310 INPUT C(1,1)
1320 PRINT "C(3,3) ?"
1330 INPUT C(3,3)
1340 PRINT "C(4,4) ?"
1350 INPUT C(4,4)
1360 PRINT "C(1,2) ?"
1370 INPUT C(1,2)
1380 PRINT "C(1,3) ?"
1390 INPUT C(1,3)
1400 PRINT "C(1,4) ?"
1410 INPUT C(1,4)
1420 PRINT "C(2,5) ?"
1430 INPUT C(2,5)
1440 C(2,2)=C(1,1)
1450 C(5,5)=C(4,4)
1460 C(2,3)=C(1,3)
1470 C(2,4)=-C(1,4)
1480 C(1,5)=-C(2,5)
1490 C(6,6)=1/2*(C(1,1)-C(1,2))
1500 C(4,6)=C(2,5)
1510 C(5,6)=-C(2,4)
1520 GOTO 1830
1530 IF System$="MON" THEN 1550
1540 IF System$<>"MON" THEN 1820
1550 PRINT "C(1,1) ?"
1560 INPUT C(1,1)
1570 PRINT "C(2,2) ?"
1580 INPUT C(2,2)
1590 PRINT "C(3,3) ?"
1600 INPUT C(3,3)
1610 PRINT "C(4,4) ?"
1620 INPUT C(4,4)
1630 PRINT "C(5,5) ?"
1640 INPUT C(5,5)
1650 PRINT "C(6,6) ?"
1660 INPUT C(6,6)
1670 PRINT "C(1,2) ?"
1680 INPUT C(1,2)
1690 PRINT "C(1,3) ?"
1700 INPUT C(1,3)
1710 PRINT "C(2,3) ?"
1720 INPUT C(2,3)
1730 PRINT "C(1,5) ?"
1740 INPUT C(1,5)
1750 PRINT "C(2,5) ?"
1760 INPUT C(2,5)
1770 PRINT "C(3,5) ?"

```

```

1780 INPUT C(3,5)
1790 PRINT "C(4,6) ?"
1800 INPUT C(4,6)
1810 GOTO 1830
1820 PRINT "INCORRECT LABEL FOR CRYSTAL SYSTEM"
1830 FOR I=1 TO 5
1840 FOR J=1 TO I
1850 C(I+1,J)=C(J,I+1)
1860 NEXT J
1870 NEXT I
1880 MAT S= INV(C)
1890 Compress=S(1,1)+S(2,2)+S(3,3)+2*(S(1,2)+S(2,3)+S(3,1))
1900 REM IF Jjjj=0 THEN 1820
1910 REM FOR I=1 TO 6
1920 REM FOR J=1 TO 6
1930 REM S(I,J)=C(I,J)
1940 REM NEXT J
1950 REM NEXT I
1960 REM MAT C= INV(S)
1970 PRINTER IS 10;WIDTH 80
1980 REM ***CALCULATE GAMMA(I,J) VALUES***
1990 M=0
2000 N=0
2010 FOR I=Tmin TO Tm STEP Dtt
2020 Theta=I*P
2030 M=M+1
2040 FOR J=Pmin TO Pm STEP Dpp
2050 Phi=J*P
2060 N=N+1
2070 REM SUBROUTINE GAMMA
2080 E11=SIN(Theta)*COS(Phi)
2090 E12=SIN(Theta)*SIN(Phi)
2100 E13=COS(Theta)
2110 E111=E11^2
2120 E122=E12^2
2130 E133=E13^2
2140 E112=E11*E12
2150 E123=E12*E13
2160 E131=E13*E11
2170 G(1,1)=E111*C(1,1)+E122*C(6,6)+E133*C(5,5)+E123*2.*C(5,6)+E131*2.*C(1,5)+E
112*2.*C(1,6)
2180 G(2,2)=E111*C(6,6)+E122*C(2,2)+E133*C(4,4)+E123*2.*C(2,4)+E131*2.*C(4,6)+E
112*2.*C(2,6)
2190 G(3,3)=E111*C(5,5)+E122*C(4,4)+E133*C(3,3)+E123*2.*C(3,4)+E131*2.*C(3,5)+E
112*2.*C(4,5)
2200 G(1,2)=E111*C(1,6)+E122*C(2,6)+E133*C(4,5)
2210 G(1,2)=G(1,2)+E123*(C(4,6)+C(2,5))+E131*(C(5,6)+C(1,4))+E112*(C(1,2)+C(6,6)
))
2220 G(2,3)=E111*C(5,6)+E122*C(2,4)+E133*C(3,4)
2230 G(2,3)=G(2,3)+E123*(C(2,3)+C(4,4))+E131*(C(4,5)+C(3,6))+E112*(C(4,6)+C(2,5)
))
2240 G(3,1)=E111*C(1,5)+E122*C(4,6)+E133*C(3,5)+E123*(C(4,5)+C(3,6))+E131*(C(1,
3)+C(5,5))
2250 G(3,1)=G(3,1)+E112*(C(5,6)+C(1,4))
2260 G(2,1)=G(1,2)
2270 G(3,2)=G(2,3)
2280 G(1,3)=G(3,1)
2290 REM ***EVALUATE ROOTS OF CUBIC EQUATION***
2300 Alph=-1.
2310 Beta=G(1,1)+G(2,2)+G(3,3)
2320 Gamm=G(1,2)^2+G(2,3)^2+G(3,1)^2-G(2,2)*G(3,3)-G(1,1)*G(3,3)-G(1,1)*G(2,2)

```

```

2330 Delta=G(1,1)*G(2,2)*G(3,3)+2.*G(1,2)*G(2,3)*G(3,1)-G(1,2)^2*G(3,3)-G(2,3)^
2*G(1,1)
2340 Delta=Delta-(G(3,1)^2)*G(2,2)
2350 H=(3.*Alph*Gamm-Beta^2)/9./Alph^2
2360 W=(Beta/100.)^3
2370 W=2.*W/27./Alph^3
2380 W=W*(100.)^3
2390 G1=W+(-Alph*Beta*Gamm/3.+Alph^2*Delta)/Alph^3
2400 Alp=-G1/2./SQR(ABS(H))^3
2410 Aalp=ABS(Alp)
2420 IF (Aalp-1.)<0 THEN GOTO 2480
2430 IF (Aalp-1.)>=0 THEN GOTO 2440
2440 Alfa=0
2450 IF (Alp)<0 THEN GOTO 2470
2460 IF (Alp)>=0 THEN GOTO 2490
2470 Alfa=(180.*1.745329E-2)
2480 Alfa=(90.*1.745329E-2)-ATN(Alp/(1.-Alp^2)^.5)
2490 Alfha=Alfa/3.
2500 Am=2.*SQR(-H)
2510 Z1=Am*COS(Alfha)
2520 Z2=Am*COS(Alfha+2.094395)
2530 Z3=Am*COS(Alfha+4.188790)
2540 X1=Z1-Beta/3./Alph
2550 X2=Z2-Beta/3./Alph
2560 X3=Z3-Beta/3./Alph
2570 X1=(SQR(X1/Rho))*1.E+1
2580 X2=(SQR(X2/Rho))*1.E+1
2590 X3=(SQR(X3/Rho))*1.E+1
2600 REM ***SORT VELOCITIES AS V(3)>V(2)>V(1)***
2610 V1=X1
2620 V2=X2
2630 V3=X3
2640 IF V1<=V2 THEN GOTO 2670
2650 V1=X2
2660 V2=X1
2670 IF V2<=V3 THEN GOTO 2710
2680 X1=V2
2690 V2=V3
2700 V3=X1
2710 IF V1<=V2 THEN GOTO 2750
2720 X1=V2
2730 V2=X1
2740 V1=X1
2750 Rav=1/V1^3+1/V2^3+1/V3^3
2760 A1(M,N)=V1
2770 A2(M,N)=V2
2780 A3(M,N)=V3
2790 A4(M,N)=Rav
2800 REM ***NUMERICAL INTEGRATION FOR V(MEAN)***
2810 Th=Theta
2820 Dw=(COS(Th)-COS(Th+Dt))*Dp
2830 IF J=Pm THEN 2880
2840 IF I=Tm THEN 2880
2850 St=St+(1./V1^3+1./V2^3)*Dw
2860 S1=S1+Dw/V3^3
2870 Sm=Sm+Rav*Dw
2880 NEXT J
2890 N=0
2900 NEXT I

```

```

2910 IF Yyyy=0 THEN 3220
2920 REM ** GENERATE PLOTTING FILES **
2930 ASSIGN @F1 TO "VEL1_DAT"
2940 R=19
2950 L=73
2960 Kk=0
2970 FOR N=1 TO L
2980 Ph=(N-1)*Dpp
2990 Kk=Kk+1
3000 OUTPUT @F1,Kk USING P1$;90,Ph,A1(R,N),A2(R,N),A3(R,N),A4(R,N)
3010 NEXT N
3020 ASSIGN @F1 TO *
3030 ASSIGN @F1 TO "VEL2_DAT"
3040 Kk=0
3050 I=1
3060 FOR N=1 TO 73
3070 Tht=(N-1)*Dtt
3080 Kk=Kk+1
3090 OUTPUT @F1,Kk USING P1$;Tht,0,A1(N,I),A2(N,I),A3(N,I),A4(N,I)
3100 NEXT N
3110 ASSIGN @F1 TO *
3120 ASSIGN @F1 TO "VEL3_DAT"
3130 Kk=0
3140 N=19
3150 FOR K=1 TO 73
3160 Tht=(K-1)*Dtt
3170 Kk=Kk+1
3180 OUTPUT @F1,Kk USING P1$;Tht,90,A1(K,N),A2(K,N),A3(K,N),A4(K,N)
3190 NEXT K
3200 ASSIGN @F1 TO *
3210 GOTO 4210
3220 Vt=((Sy/12.56637*St)/2.)^(-.3333333)
3230 V1=(Sy/12.56637*S1)^(-.3333333)
3240 Vm=((Sy/12.56637*Sm)/3.)^(-.3333333)
3250 Debye=Hk*(3.*Iq*Avog*Rho/12.56637/Formwt)^.3333333*Vm*100000
3260 REM ***PRINTOUT OF C(I,J)'S AND VELOCITIES***
3270 PRINT Name$,Day$
3280 PRINT
3290 PRINT USING "10A,2X,DD.DDDD,2X,7A";"DENSITY---",Rho,"GM/CM^3"
3300 PRINT
3310 PRINT USING "16A,2X,DDD.DDD,2X,7A";"FORMULA WEIGHT--",Formwt,"GM/MOLE"
3320 PRINT
3330 PRINT USING "30A,2X,DDD";"NUMBER OF ATOMS IN MOLECULE--",Iq
3340 PRINT
3350 PRINT "INTEGRATION IN THETA AND PHI"
3360 PRINT
3370 PRINT USING "34A,X,DD.DDDD,X,6A";"MEAN VELOCITY OF TRANSVERSE WAVES--",Vt,
"KM/SEC"
3380 PRINT
3390 PRINT USING "35A,X,DD.DDDD,X,6A";"MEAN VELOCITY OF LONGITUDINAL WAVE--",V1
,"KM/SEC"
3400 PRINT
3410 PRINT USING "19A,X,DD.DDDD,X,6A";"MEAN SOUND VELOCITY--",Vm,"KM/SEC"
3420 PRINT
3430 PRINT USING "19A,X,DDDD.D,X,6A";"DEBYE TEMPERATURE--",Debye,"KELVIN"
3440 PRINT
3450 PRINT USING "17A,X,DD.DD,X,6A";"COMPRESSIBILITY--",Compress,"1/MBAR"
3460 PRINT
3470 PRINT "THE C(I,J) IN MEGABARS"

```

```

3480 PRINT
3490 PRINT USING "XX,DD.DDDD";C(1,1),C(1,2),C(1,3),C(1,4),C(1,5),C(1,6)
3500 PRINT USING "11X,4(DD.4D,XX),DD.4D";C(2,2),C(2,3),C(2,4),C(2,5),C(2,6)
3510 PRINT USING "20X,3(DD.4D,XX),DD.DDDD";C(3,3),C(3,4),C(3,5),C(3,6)
3520 PRINT USING "29X,DD.DDDD,XX,DD.DDDD,XX,DD.DDDD";C(4,4),C(4,5),C(4,6)
3530 PRINT USING "38X,DD.DDDD,XX,DD.DDDD";C(5,5),C(5,6)
3540 PRINT USING "47X,DD.DDDD";C(6,6)
3550 PRINT
3560 PRINT
3570 PRINT "THE S(I,J) IN RECIPROCAL MEGABARS"
3580 PRINT
3590 PRINT USING "XX,DD.DDDD";S(1,1),S(1,2),S(1,3),S(1,4),S(1,5),S(1,6)
3600 PRINT USING "11X,4(DD.4D,XX),DD.4D";S(2,2),S(2,3),S(2,4),S(2,5),C(2,6)
3610 PRINT USING "20X,3(DD.4D,XX),DD.DDDD";S(3,3),S(3,4),S(3,5),S(3,6)
3620 PRINT USING "29X,DD.DDDD,XX,DD.DDDD,XX,DD.DDDD";S(4,4),S(4,5),S(4,6)
3630 PRINT USING "38X,DD.DDDD,XX,DD.DDDD";S(5,5),S(5,6)
3640 PRINT USING "47X,DD.DDDD";S(6,6)
3650 PRINT
3660 PRINT
3670 PRINT
3680 PRINT "SOUND WAVE VELOCITY (IN KM/SEC) AS A FUNCTION OF THETA AND PHI IN T
HE XY PLANE"
3690 PRINT
3700 PRINT "
                                **TRANSVERSE**    LONGITUDINAL    RECIPROCAL AVERAG
E"
3710 PRINT "THETA      PHI      V1      V2      V3      (1/V1^3+1/V2^3+1/V3^
3)"
3720 PRINT
3730 PRINT
3740 IF Tm<>180 THEN 3780
3750 IF Dtt=5 THEN K=19
3760 IF Dtt=10 THEN K=10
3770 GOTO 3790
3780 K=Tm/Dtt+1
3790 R=K
3800 L=Pm/Dpp+1
3810 FOR N=1 TO L
3820 Ph=(N-1)*Dpp
3830 IMAGE DDD.D,XXX,DDD.D,XXX,DD.DDD,XXX,DD.DDD,XXXXXX,DD.DDD,9X,.DDDDDE
3840 PRINT USING 3830;90,Ph,A1(R,N),A2(R,N),A3(R,N),A4(R,N)
3850 NEXT N
3860 PRINT
3870 PRINT
3880 PRINT
3890 PRINT
3900 PRINT "SOUND WAVE VELOCITY (IN KM/SEC) AS A FUNCTION OF THETA AND PHI IN T
HE XZ PLANE"
3910 PRINT
3920 PRINT "
                                **TRANSVERSE**    LONGITUDINAL    RECIPROCAL AVERAG
E"
3930 PRINT "THETA      PHI      V1      V2      V3      (1/V1^3+1/V2^3+1/V3^
3)"
3940 PRINT
3950 PRINT
3960 I=1
3970 K=Tm/Dtt+1
3980 FOR N=1 TO K
3990 Tht=(N-1)*Dtt
4000 PRINT USING 3830;Tht,Pmin,A1(N,I),A2(N,I),A3(N,I),A4(N,I)
4010 NEXT N
4020 PRINT
4030 PRINT

```

```

4040 PRINT "SOUND WAVE VELOCITY (IN KM/SEC) AS A FUNCTION OF THETA AND PHI IN T
HE YZ PLANE"
4050 PRINT "THETA      PHI      V1      V2      V3      (1/V1^3+1/V2^3+1/V3^
3)"
4060 PRINT
4070 PRINT
4080 IF Pm<>180 THEN 4120
4090 IF Dpp=5 THEN L=19
4100 IF Dpp=10 THEN L=10
4110 GOTO 4160
4120 IF Pm<>120 THEN 4160
4130 IF Dpp=5 THEN L=19
4140 IF Dpp=10 THEN L=10
4150 GOTO 4160
4160 N=L
4170 FOR K=1 TO K
4180 Tht=(K-1)*Dtt
4190 PRINT USING 3830;Tht,90,A1(K,N),A2(K,N),A3(K,N),A4(K,N)
4200 NEXT K
4210 PRINTER IS 1;WIDTH 50
4220 END

```